
[bookmark: clause_CommOps_CatchOp][bookmark: _Toc514234969]22.3.6	The Catch operation
The catch operation is used to catch exceptions.
Syntactical Structure
( ObjectReference | any port | any from PortArrayRef ) "." catch 
[ "(" ( Signature [ "," TemplateInstance ] ) | TimeoutKeyword ")" ] 
[ from Address ]
[ "->" [ value ( ValueRef| 
                 ( "(" { ValueRef[ ":=" [ @decoded [ "(" Expression ")" ] ]												FieldOrTypeReference ][","] } ")" )
                ) ]
       [ sender ValueRef] 
       [ @index value ValueRef] ]

NOTE 1:	Address may be an AddressRef, a list of AddressRef-s or "any component".
Semantic Description
The catch operation is used to catch exceptions raised by a test component or the SUT as a reaction to a procedure call. Exceptions are specified as types and thus, can be treated like messages, e.g. templates can be used to distinguish between different values of the same exception type. If a Signature is given in the parameter list, it is possible to omit the TemplateInstance part if the catch operation shall match any exception value of any of the exception types declared in the definition of the referenced Signature.
The catch operation removes the top exception from the associated incoming port queue if, and only if, that top exception satisfies all the matching criteria associated with the catch operation. 
A catch operation may be restricted to a certain communication partner in case of one-to-many connections. This restriction shall be denoted by using the from keyword followed by a specification of an address or component reference, a list of address or component references or any component.
NOTE 2:	The one-to-one connection is considered to be a simple case of the one-to-many connections and allows the usage of the from-clause.
The (optional) redirection part of the catch operation comprises of storing the exception value and/or one or more parts of it and the retrieval of the address of the calling component. The keyword value is used to retrieve the value of an exception and/or the parts of it and the keyword sender is used when it is required to retrieve the address of the sender. 
When assigning individual fields of an exception, encoded payload fields can be decoded prior to assignment using the @decoded modifier. In this case, the referenced field on the right hand sided of the assignment shall be one of the bitstring, hexstring, octetstring, charstring or universal charstring types. It shall be decoded into a value of the same type as the variable on the left hand side of the assignment. Failure of this decoding shall cause a test case error. In case the referenced field is of the universal charstring type, the @decoded clause can contain an optional parameter defining the encoding format. The parameter shall be of the charstring type and it shall contain one of the strings allowed for the decvalue_unichar function (specified in clause C.5.4). Any other value shall cause an error. In case the referenced field is not a universal charstring, the optional parameter shall not be present.
The catch operation may be part of the response and exception handling part of a call operation or be used to determine an alternative in an alt statement. If the catch operation is used in the accepting part of a call operation, the information about port name and signature reference to indicate the procedure that raised the exception is redundant, because this information follows from the call operation. However, for readability reasons (e.g. in case of complex call statements) this information shall be repeated.
The Timeout exception
There is one special timeout exception that can be caught by the catch operation. The timeout exception is an emergency exit for cases where a called procedure neither replies nor raises an exception within a predetermined time (see clause 22.3.1).
Catch any exception
A catch operation with no argument list allows any valid exception to be caught. The most general case is without using the from keyword. CatchAnyException will also catch the timeout exception.
Catch any exception for specific signature
A catch operation using only a Signature reference in the argument list allows any valid exception for that signature to be caught. 
Catch on any port
To catch an exception on any port use the any keyword.
Catch on any port from a port array
To catch an exception on any port from a specific port array, indices use the any from PortArrayRef syntax where PortArrayRef shall be a reference to a  port array identifier. It is also possible to store the index of a port in a single-dimensional port array at which the operation was successful to a variable of type integer or, in case of multi‑dimensional port arrays the index of the successful port to an integer array or record of integer variable. When checking the port array for matching exceptions, the port indices to be checked are iterated from lowest to highest. If the port array is multi-dimensional, then the ports are iterated over from innermost to outermost array dimension from lowest to highest index for each dimension, e.g. [0][0], [0][1], [1][0], [1][1]. The first port which matches all the criteria will cause the operation to be successful even if other ports in the array would also meet the criteria.
The catch on any port from a port array operation cannot be used to catch a call timeout.
Restrictions
In addition to the general static rules of TTCN‑3 given in clause 5 and shown in table 16, the following restrictions apply:
1. The catch operation shall only be used at procedure-based ports. The type of the caught exception shall be specified in the signature of the procedure that raised the exception.
1. The type definition of the port shall include in its out or inout list the name of the procedure to which the exception belongs.
a) No binding of the incoming values to the terms of the expression or to the template shall occur. The assignment of the exception values to variables shall be made in the assignment part of the catch operation.
b) Catching timeout exceptions shall be restricted to the exception handling part of a call. No further matching criteria (including a from part) and no assignment part is allowed for a catch operation that handles a timeout exception.
c) Exception values accepted by catch any exception shall not be assigned to a variable, i.e. the value clause shall not be present.
d) If CatchAnyException is used in the response and exception handling part of a call operation, it shall only treat exceptions raised by the procedure invoked by the call operation.
e) All AddressRef items in the from clause and all ValueRef items in the sender clause shall be of type address, component or of the address type bound to the port type (see clause 6.2.9) of the port instance referenced in the catch operation. No AddressRef in the from clause shall contain the special value null at the time of the operation.
f) The PortArrayRef shall be a reference to a completely initialized port array.
g) The index redirection shall only be used when the operation is used on an any from port array construct.
h) If the index redirection is used for single-dimensional arrays, the type of the integer variable shall allow storing the highest index of the respective port array.
i) If the index redirection is used for multi-dimensional arrays, the size of the integer array or record of integer type shall exactly be the same as the dimension of the respective port array, and the its type shall allow storing the highest index (from all dimensions) of the port array.
j) If a variable referenced in the value, sender or @index clause is a lazy or fuzzy variable, the expression assigned to this variable is equal to the result produced by the catch operation, i.e. later evaluation of the lazy or fuzzy variable does not lead to repeated invocation of the catch operation.
k) If the catch operation contains both from and sender clause, the variable or parameter referenced in the sender clause shall be type compatible with the template in the from clause. If the operation contains a sender clause but no from clause, the sender shall be type compatible with the variable or parameter referenced in the sender clause.
NOTE 3:	An error due to a type mismatch may happen if the types in the receive part are not compatible to the types in the assignment part or, if the from clause is missing, but the type of the sender can be determined and it is not type compatible with the type in the sender clause.
l) When assigning implicitly decoded exception fields (by using the @decoded modifier) in cases where the value or template to be matched uses the MatchDecodedContent (decmatch) matching for the parameter to be stored, the type of the template in the MatchDecodedContent matching shall be type-compatible to the type of the variable the decoded field is stored into.
m) The referenced value associated with Ref or the return type associated with FunctionInstance followed by the catch keyword, shall be of a port type.
n) If no TemplateInstance is provided in the parameter list, then also no value clause shall be present in the redirection part.
o) 
Examples
EXAMPLE 1:	Basic catch
	myPort.catch(MyProc, integer: v_myVar);	   // Catches an integer exception of value
											   // v_myVar raised by MyProc at port myPort.

	myPort.catch(MyProc, v_myVar);			   // Is an alternative to the previous example.

	myPort.catch(MyProc, v_a<v_b);			   // Catches a boolean exception

	myPort.catch(MyProc, MyType:{5, v_myVar}); // In-line template definition of an exception value.

	myPort.catch(MyProc, charstring:"Hello")from myPeer;	// Catches "Hello" exception from myPeer

EXAMPLE 2:	Catch with storing value and/or sender in variables
	myPort.catch(MyProc, MyType:?) from myPartner -> value v_myVar;
	// Catches an exception from myPartner and assigns its value to v_myVar.

	myPort.catch(MyProc, s_myTemplate(5)) -> value v_myVarTwo sender myPeer;
	// Catches an exception, assigns its value to v_myVarTwo and retrieves the
	// address of the sender.

	myPort.catch(MyProc, s_myTemplate(5)) -> value (v_myVarThree:= f1)
										   	 sender myPeer;
	// Catches an exception, assigns the value of its field f1 to v_myVarThree and retrieves the
	// address of the sender.

	// Handling encoded exception payload:

	type MyException record {
	  ...
	}
	type CommonException record {
	  integer		exceptionId,
	  octetstring	payload
	}

	signature S() exception (CommonException);
	...

	var MyException v_myVar;

	myPort.catch (S, CommonException:{exceptionId := 25, payload := decmatch MyException:? }) 										-> value (v_myVar := @decoded payload);
	// The encoded payload field of the caught exception is decoded and matched with m_excTemplate;
    // if the matching is successful the decoded payload is stored in v_myVar.

EXAMPLE 3:	The Timeout exception
	myPort.call(MyProc:{5, v_myVar}, 20E-3) {
	  [] myPort.getreply(MyProc:{?, ?}) { }
	  [] myPort.catch(timeout) {				// timeout exception after 20ms
			setverdict(fail);
			stop;
	     }
	}

EXAMPLE 4:	Catch any exception
	myPort.catch;

	myPort.catch from myPartner;

	myPort.catch -> sender v_mySenderVar;

	myPort.catch(MyProc); // catch any exception raised by procedure MyProc

EXAMPLE 5:	Catch on any port
	any port.catch;

EXAMPLE 6:	Catch on any port from port array
    type port MyPort procedure { inout MyProc }
    type component MyComponent {
      port MyPort p[10][10];
    }
    var integer v_i[2];
    any from p.catch(MyProc, MyType:?) -> @index value v_i; 
    // Catching an incoming exception of type MyType on any port in the port array p and
    // storing the index of the port on which the matching was successful first


[bookmark: _Toc514235049]A.1.6.4.2	Port operations
[bookmark: TCommunicationStatements]CommunicationStatements ::= SendStatement | 
                                 CallStatement | 
                                 ReplyStatement | 
                                 RaiseStatement | 
                                 ReceiveStatement | 
                                 TriggerStatement | 
                                 GetCallStatement | 
                                 GetReplyStatement | 
                                 CatchStatement | 
                                 CheckStatement | 
                                 ClearStatement | 
                                 StartStatement | 
                                 StopStatement | 
                                 HaltStatement | 
                                 CheckStateStatement 
[bookmark: TSendStatement]SendStatement ::= ObjectReference Dot PortSendOp 
[bookmark: TPortSendOp]PortSendOp ::= SendOpKeyword "(" TemplateInstance")" [ToClause] 
[bookmark: TSendOpKeyword]SendOpKeyword ::= "send" 
[bookmark: TToClause]ToClause ::= ToKeyword (TemplateInstance| 
                             AddressRefList | 
                             AllKeyword ComponentKeyword 
                            ) 
[bookmark: TAddressRefList]AddressRefList ::= "(" TemplateInstance{"," TemplateInstance} ")" 
[bookmark: TToKeyword]ToKeyword ::= "to" 
[bookmark: TCallStatement]CallStatement ::= ObjectReference Dot PortCallOp [PortCallBody] 
[bookmark: TPortCallOp]PortCallOp ::= CallOpKeyword "(" CallParameters ")" [ToClause] 
[bookmark: TCallOpKeyword]CallOpKeyword ::= "call" 
[bookmark: TCallParameters]CallParameters ::= TemplateInstance ["," CallTimerValue] 
[bookmark: TCallTimerValue]CallTimerValue ::= Expression | NowaitKeyword 
[bookmark: TNowaitKeyword]NowaitKeyword ::= "nowait" 
[bookmark: TPortCallBody]PortCallBody ::= "{" CallBodyStatementList "}" 
[bookmark: TCallBodyStatementList]CallBodyStatementList ::= {CallBodyStatement [SemiColon]}+ 
[bookmark: TCallBodyStatement]CallBodyStatement ::= CallBodyGuard StatementBlock 
[bookmark: TCallBodyGuard]CallBodyGuard ::= AltGuardChar CallBodyOps 
[bookmark: TCallBodyOps]CallBodyOps ::= GetReplyStatement | CatchStatement 
[bookmark: TReplyStatement]ReplyStatement ::= ObjectReference Dot PortReplyOp 
[bookmark: TPortReplyOp]PortReplyOp ::= ReplyKeyword "(" TemplateInstance [ReplyValue] ")" [ToClause] 
[bookmark: TReplyKeyword]ReplyKeyword ::= "reply" 
[bookmark: TReplyValue]ReplyValue ::= ValueKeyword TemplateBody
/* STATIC SEMANTICS - TemplateBody shall be type compatible with the return type. It shall evaluate to a value or template (literal or template instance)  conforming to the template(value) restriction. */
[bookmark: TRaiseStatement]RaiseStatement ::= ObjectReference Dot PortRaiseOp 
[bookmark: TPortRaiseOp]PortRaiseOp ::= RaiseKeyword "(" Signature "," TemplateInstance")"   
                     [ToClause] 
[bookmark: TRaiseKeyword]RaiseKeyword ::= "raise" 
[bookmark: TReceiveStatement]ReceiveStatement ::= PortOrAny Dot PortReceiveOp 
[bookmark: TPortOrAny]PortOrAny ::= ObjectReference | (AnyKeyword (PortKeyword | FromKeyword ValueRef)) 
[bookmark: TPortReceiveOp]PortReceiveOp ::= ReceiveOpKeyword ["("TemplateInstance")"] [FromClause] [PortRedirect] 
[bookmark: TReceiveOpKeyword]ReceiveOpKeyword ::= "receive" 
[bookmark: TFromClause]FromClause ::= FromKeyword (TemplateInstance | 
                                 AddressRefList | 
                                 AnyKeyword ComponentKeyword 
                                ) 
[bookmark: TFromKeyword]FromKeyword ::= "from" 
[bookmark: TPortRedirect]PortRedirect ::= PortRedirectSymbol ((ValueSpec [SenderSpec] [IndexSpec]) | 
                                         (SenderSpec [IndexSpec]) | 
                                          IndexSpec 
                                         ) 
[bookmark: TPortRedirectSymbol]PortRedirectSymbol ::= "->" 
[bookmark: TValueSpec]ValueSpec ::= ValueKeyword (ValueRef | ("(" SingleValueSpec {"," SingleValueSpec} ")")) 
[bookmark: TSingleValueSpec]SingleValueSpec ::= ValueRef [AssignmentChar [ DecodedModifier ["(" [Expression] ")"] ]
                                                    FieldReference ExtendedFieldReference] 

/*STATIC SEMANTICS – FieldReference shall not be ParRef and ExtendedFieldReference shall not be TypeDefIdentifier*/ 
[bookmark: TValueKeyword]ValueKeyword ::= "value" 
[bookmark: TSenderSpec]SenderSpec ::= SenderKeyword ValueRef
[bookmark: TSenderKeyword]SenderKeyword ::= "sender" 
[bookmark: TTriggerStatement]TriggerStatement ::= PortOrAny Dot PortTriggerOp 
[bookmark: TPortTriggerOp]PortTriggerOp ::= TriggerOpKeyword ["(" TemplateInstance  ")"] [FromClause]   
                       [PortRedirect] 
[bookmark: TTriggerOpKeyword]TriggerOpKeyword ::= "trigger" 
[bookmark: TGetCallStatement]GetCallStatement ::= PortOrAny Dot PortGetCallOp 
[bookmark: TPortGetCallOp]PortGetCallOp ::= GetCallOpKeyword ["(" TemplateInstance ")"] [FromClause]   
                       [PortRedirectWithParam] 
[bookmark: TGetCallOpKeyword]GetCallOpKeyword ::= "getcall" 
[bookmark: TPortRedirectWithParam]PortRedirectWithParam ::= PortRedirectSymbol RedirectWithParamSpec 
[bookmark: TRedirectWithParamSpec]RedirectWithParamSpec ::= (ParamSpec [SenderSpec] [IndexSpec]) | 
                               (SenderSpec [IndexSpec]) | 
                               IndexSpec 
[bookmark: TParamSpec]ParamSpec ::= ParamKeyword ParamAssignmentList 
[bookmark: TParamKeyword]ParamKeyword ::= "param" 
[bookmark: TParamAssignmentList]ParamAssignmentList ::= "(" (AssignmentList | VariableList) ")" 
[bookmark: TAssignmentList]AssignmentList ::= VariableAssignment {"," VariableAssignment} 
[bookmark: TVariableAssignment]VariableAssignment ::= ValueRef AssignmentChar [ DecodedModifier ["(" Expression] ")"]
                                                      Identifier 
[bookmark: TVariableList]VariableList ::= VariableEntry {"," VariableEntry} 
[bookmark: TVariableEntry]VariableEntry ::= ValueRef | Minus 
[bookmark: TGetReplyStatement]GetReplyStatement ::= PortOrAny Dot PortGetReplyOp 
[bookmark: TPortGetReplyOp]PortGetReplyOp ::= GetReplyOpKeyword ["(" TemplateInstance [ValueMatchSpec]   
                                           ")"] [FromClause] [PortRedirectWithValueAndParam] 
[bookmark: TPortRedirectWithValueAndParam]PortRedirectWithValueAndParam ::= PortRedirectSymbol RedirectWithValueAndParamSpec 
[bookmark: TRedirectWithValueAndParamSpec]RedirectWithValueAndParamSpec ::= (ValueSpec [ParamSpec] [SenderSpec]   
                                        [IndexSpec]) | RedirectWithParamSpec 
[bookmark: TGetReplyOpKeyword]GetReplyOpKeyword ::= "getreply" 
[bookmark: TValueMatchSpec]ValueMatchSpec ::= ValueKeyword TemplateInstance
[bookmark: TCheckStatement]CheckStatement ::= PortOrAny Dot PortCheckOp 
[bookmark: TPortCheckOp]PortCheckOp ::= CheckOpKeyword ["(" CheckParameter ")"] 
[bookmark: TCheckOpKeyword]CheckOpKeyword ::= "check" 
[bookmark: TCheckParameter]CheckParameter ::= CheckPortOpsPresent | 
                        FromClausePresent | 
                        RedirectPresent 
[bookmark: TFromClausePresent]FromClausePresent ::= FromClause [PortRedirectSymbol ((SenderSpec   
                                                            [IndexSpec]) | 
                                                           IndexSpec)] 
[bookmark: TRedirectPresent]RedirectPresent ::= PortRedirectSymbol ((SenderSpec [IndexSpec]) | 
                                             IndexSpec) 
[bookmark: TCheckPortOpsPresent]CheckPortOpsPresent ::= PortReceiveOp | 
                             PortGetCallOp | 
                             PortGetReplyOp | 
                             PortCatchOp 
[bookmark: TCatchStatement]CatchStatement ::= PortOrAny Dot PortCatchOp 
[bookmark: TPortCatchOp]PortCatchOp ::= CatchOpKeyword ["(" CatchOpParameter ")"] [FromClause] [PortRedirect] 
[bookmark: TCatchOpKeyword]CatchOpKeyword ::= "catch" 
[bookmark: TCatchOpParameter]CatchOpParameter ::= Signature [ "," TemplateInstance ] | TimeoutKeyword 
[bookmark: TClearStatement]ClearStatement ::= PortOrAll Dot ClearOpKeyword 
[bookmark: TPortOrAll]PortOrAll ::= ObjectReference | AllKeyword PortKeyword 
[bookmark: TClearOpKeyword]ClearOpKeyword ::= "clear" 
[bookmark: TStartStatement]StartStatement ::= PortOrAll Dot StartKeyword 
[bookmark: TStopStatement]StopStatement ::= PortOrAll Dot StopKeyword 
[bookmark: TStopKeyword]StopKeyword ::= "stop" 
[bookmark: THaltStatement]HaltStatement ::= PortOrAll Dot HaltKeyword 
[bookmark: THaltKeyword]HaltKeyword ::= "halt" 
[bookmark: TAnyKeyword]AnyKeyword ::= "any" 
[bookmark: TCheckStateStatement]CheckStateStatement ::= PortOrAllAny Dot CheckStateKeyword "(" SingleExpression   
                             ")" 
[bookmark: TPortOrAllAny]PortOrAllAny ::= PortOrAll | AnyKeyword PortKeyword 
[bookmark: TCheckStateKeyword]CheckStateKeyword ::= "checkstate" 

